
COMMUTATORE D’ANTENNA AUTOMATICO

Il commutatore d’antenna automatico si basa sul PIC 16F877 della Microchip
Il codice è scritto in mikroBasic per facilitarne l’interpretazione.

Per mezzo del CAT basato sulla porta seriale RS232 a bordo delle nostre radio, 
con opportuni accorgimenti è possibile interrogare il VFO A o B con l’invio di  
stringhe
 
Ad esempio, con una radio Kenwood TS870, inviando la stringa “FA;” la radio 
risponde ad esempio con “FA00007100000;” che corrisponde alla frequenza di 
7.100,000MHz ovvero banda dei 40Mt del VFO A

Notare che la stringa ad esempio, comando “FA;” termina con un “;” come per 
la risposta inviata dalla radio

Il “;” è il carattere che indica alla CPU della radio che la stringa è terminata
pertanto tutto quello che viene dopo il “;” viene ignorato
Per le stringhe di comando fare sempre riferimento al manuale della radio

Una volta stabilita la comunicazione non resta che fare dei confronti tra 
frequenza letta e relè che si vuole attivare per la specifica banda

Ad esempio:

' BANDA 80MT----------------------------

   IF (strncmp("FA00003",uart_rd,7)) THEN
   
      Portd.1 = 0

   ELSE

      Portd.1 = 1
  END IF

La parte di codice visibile sopra viene utilizzata per comandare il relè 1 della 
porta D dedicato agli 80Mt, se si verifica la seguente condizione:

Se (IF) la comparazione della stringa (strncmp) inviata dalla radio (uart_rd), 
nei primi 7 caratteri è composta da “FA00003”, la funzione (strncmp) restituirà 
un valore di 0

Allora (THEN) disattiva l’uscita 1 della porta D se il valore restituito è diverso 
da 0

Altrimenti (ELSE) se il valore restituito è uguale a 0, attiva l’uscita 1 della porta
D dedicata al relè per l’antenna degli 80Mt 

Questa condizione è utilizzabile per tutte le bande di nostro interesse con 
opportune modifiche, dove si vuole comandare un relè dedicato per l’inserzione
dell’antenna

Inoltre è possibile inviare dei comandi alla radio come un  cambio banda

Ad esempio:

Selezione:

 IF portb.2 = 1 THEN
               Goto Banda40
            END IF

'---------------------------------------------------
   Banda40:

   IF(UART1_Tx_Idle = 1) THEN
   UART1_Write_Text("FA00007000000;")
   SPI_Lcd_Out(3,1,"OK 40Mt")
   Goto Selezione
   END IF

Per mezzo della funzione “UART1_Write_Text(“FA00007000000;”)”
è possibile impostare il VFO A sulla banda dei 40Mt alla frequenza di 
7.000,000MHz con la semplice pressione di un pulsante che in questo caso è 
collegato alla porta B ingresso 2

Le informazioni di stato sono visibili per mezzo di un display LCD a quattro 
linee orizzontali per venti caratteri per linea

Ad esempio:

SPI_Lcd_Out(3,1,”OK 40 Mt”)

Dove “SPI_Lcd” è il tipo di comunicazione tra il PIC e il display, “Out” è la 
funzione di scrittura su display, “3” indica che si vuole scrivere alla terza riga, 
“1” indica che il testo deve partire dal primo carattere verso destra

Il comando “Goto” serve per passare incondizionatamente a un’etichetta ad 
esempio “Banda40” e fare eseguire il codice al suo interno

Codi utilizzato per il commutatore d’antenna:

program Prova

dim uart_rd as string[20]

'Definizione dei pin per display LCD SPI
dim
SPExpanderRST as sbit at RC0_bit
SPExpanderCS as sbit at RC1_bit
SPExpanderRST_Direction as sbit at TRISC0_bit
SPExpanderCS_Direction as sbit at TRISC1_bit
'-----------------------------------------------
main:

'Reset delle porte------------------------------

 PORTA = 0
 PORTB = 0
 PORTC = 0
 PORTD = 0
 PORTE = 0
 
 'Configurazione delle porte---------------------

  TRISA = 1 ' Input
  TRISB = 1 ' Input
  TRISC = 0 ' Output
  TRISD = 0 ' Output
  TRISE = 0 ' Output
  
 'Configura le porte come digitali
  ADCON1 = 0x06

'Inizializzazione del display LCD SPI

SPI1_Init()                  'Inizializzazione del modulo SPI
SPI_Lcd_Config(0)            'Inizializzazione dell'interfaccia SPI LCD
SPI_Lcd_Cmd(_LCD_CLEAR)      'Cancella display
SPI_Lcd_Cmd(_LCD_CURSOR_OFF) 'Disattiva cursore sul display
'
SPI_Lcd_Out(1,1, "------ HB9TTK -----")
Delay_ms(600)

SPI_Lcd_Out(2,1, "AUTOMATIC ANTENNA")

Delay_ms(600)
SPI_Lcd_Out(3,1, "SWITCH")
Delay_ms(600)
SPI_Lcd_Out(4,1, "VERSION: 1.0")
Delay_ms(2000)
SPI_Lcd_Cmd(_LCD_CLEAR)
Delay_ms(600)
SPI_Lcd_Out(1,1, "VFO A MHz:")

'---------------------------------------------------
'NOTE:
' portc.2 Output come RTS sulla piastra Comandato dalla MCU
' portb.0 Input come CTS sulla piastra Comandato dalla radio
'---------------------------------------------------

'Inizializzazione seriale
UART1_Init(2400)
delay_ms(100)

'---------------------------------------------------
 while TRUE

 Selezione:
 
            IF portb.1 = 1 THEN
               Goto Banda20
            END IF
              
            IF portb.2 = 1 THEN
               Goto Banda40
            END IF
            
            IF portb.3 = 1 THEN
            portd.6 =  not portd.6
            'ELSE
            'portd.6 = 0
            END IF

            IF (portb.0 = 0) 
            AND (portb.1 = 0) 
            AND (portb.2 = 0) 
            AND (portb.3 = 0)  THEN

             'Delay_ms(500) 'Tempo di Pausa se richiesto
               Portc.2 = 1
               Goto ScriviComando
             END IF
            

wend
'---------------------------------------------------
   Banda40:
   IF(UART1_Tx_Idle = 1) THEN
   UART1_Write_Text("FA00007000000;")
   SPI_Lcd_Out(3,1,"OK 40Mt")
   Goto Selezione
   END IF
'---------------------------------------------------
   Banda20:
   
   IF(UART1_Tx_Idle = 1) THEN
   UART1_Write_Text("FA00014000000;")
   SPI_Lcd_Out(3,1,"OK 20Mt")
   Goto Selezione
   END IF
 '---------------------------------------------------
   ScriviComando:

   IF(UART1_Tx_Idle = 1) THEN
   UART1_Write_Text("FA;")
   Goto ScriviDisplay
   END IF
 
   ScriviDisplay:
   Portc.2 = 0
   UART1_Read_Text(uart_rd,";",14)
   SPI_Lcd_Out(2,1,uart_rd)
   Goto Confronta

'---------------------------------------------------
   Confronta:

  ' BANDA 160MT----------------------------
  
   IF (strncmp("FA00001",uart_rd,7)) THEN

      Portd.0 = 0
   ELSE
      Portd.0 = 1
   END IF
   
  ' BANDA 80MT----------------------------

   IF (strncmp("FA00003",uart_rd,7)) THEN
   
      Portd.1 = 0
   ELSE
      Portd.1 = 1
   END IF
   
  ' BANDA 40MT----------------------------

   IF (strncmp("FA00007",uart_rd,7)) THEN

      Portd.2 = 0
   ELSE
      Portd.2 = 1
   END IF

  ' BANDA 20MT----------------------------

   IF (strncmp("FA00014",uart_rd,7)) THEN

      Portd.3 = 0
   ELSE
      Portd.3 = 1
   END IF

  ' BANDA 15MT----------------------------

   IF (strncmp("FA00021",uart_rd,7)) THEN

      Portd.4 = 0
   ELSE
      Portd.4 = 1
   END IF
   
  ' BANDA 10MT----------------------------

   IF (strncmp("FA00028",uart_rd,7)) THEN

      Portd.5 = 0
   ELSE
      Portd.5 = 1
   END If
   
   
' DISPLAY RELE SELEZIONATO-------------------
  
  IF Portd.0 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 1")
  END IF
  
  IF Portd.1 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 2")
  END IF
  
  IF Portd.2 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 3")
  END IF
  
  IF Portd.3 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 4")
  END IF
  
  IF Portd.4 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 5")
  END IF
  
  IF Portd.5 = 1 THEN
  SPI_Lcd_Out(4,1,"RELE ANTENNA 6")

  END IF
  
  IF (not Portd.0 = 1)
  AND(not Portd.1 = 1)
  AND(not Portd.2 = 1)
  AND(not Portd.3 = 1)
  AND(not Portd.4 = 1)
  AND(not Portd.5 = 1)  THEN

  SPI_Lcd_Out(4,1,"RELE ANTENNA -")
  END IF
   Goto Selezione

end.

Fine del codice

Il materiale per lo sviluppo di questa applicazione è stato acquistato da 
https://www.mikroe.com/ dove si possono trovare ulteriori informazioni

- Board di sviluppo: EasyPic6

- Compilatore: mikroBasic PRO for PIC
scaricabile gratuitamente
Il compilatore è possibile utilizzarlo in versione demo senza limitazioni nelle 
sue funzioni ma solo sulle dimensioni del codice

